JAPLA研究会資料 2016/8/6

超幾何関数(Hypergeometric Function)とＪプログラム

－誤差関数、正規分布関数、累積密度関数を目指して－

西川　利男

１．超幾何関数とは－その位置づけ

超幾何関数(Hypergeometric Function)という名を聞いたことがあるだろうか。あまりにも高度で、特殊なものと思われるかもしれないが、そんなことはない。ここではＪプログラミングの練習課題として、とりあげて見よう。

C. F. Gaussに始まり、その後多くの数学者により、さまざまな展開がなされている。

１．１　超幾何関数の定義[1][2]

・Gaussの超幾何関数

$$F(\alpha, \beta, \gamma; x) = 1 + \frac{\alpha \cdot \beta}{\gamma \cdot 2!} x + \frac{\alpha(\alpha+1) \cdot \beta(\beta+1)}{\gamma(\gamma+1) \cdot 3!} x^2 + \frac{\alpha(\alpha+1)(\alpha+2) \cdot \beta(\beta+1)(\beta+2)}{\gamma(\gamma+1)(\gamma+2) \cdot 4!} x^3 + \cdots$$

・合流形超幾何関数(Kummerの超幾何関数)　(後述4.に)

$$M(\alpha, \gamma; x) = 1 + \frac{\alpha}{\gamma \cdot 2!} x + \frac{\alpha(\alpha+1)}{\gamma(\gamma+1) \cdot 3!} x^2 + \frac{\alpha(\alpha+1)(\alpha+2)}{\gamma(\gamma+1)(\gamma+2) \cdot 4!} x^3 + \cdots$$

１．２　その位置づけ

いろいろな数学の関数がある。あるものは増加したり、また減少したり、あるいはいろいろと変動したり、繰り返し振動したりといろいろである。

まず初等関数と呼ばれる x^2, x^3, $\sqrt{x}$, ln x, log x, sin x, cos x,..がある。また、面積を求めたり、微分方程式の解を表すのに積分による関数表示がある。さらに、たとえば楕円の長さなどに関する楕円関数、あるいはベッセル関数などといった特殊関数と呼ばれるものもある。

一方、どんな関数でも係数をいろいろと決めることによりべき乗の多項式の級数の和として表される。そのひとつがTaylor級数である。さらにその係数をもっと拡張したものはいろいろ多方面に対応できる。それが万能の多項式級数＝超幾何関数である。また、これにより数値的取り扱いが可能になるという数値計算の強力な道具になる。

[1] 高木貞治「解析概論」改定第３版　p.188, 岩波書店(1986).

[2] 森口繁一、宇田川銈一、一松信「数学公式Ⅲ」p.55-81, 岩波書店(1977).

２．超幾何関数の利用例－自然対数の値を求める

自然対数の超幾何関数による表示

$$\frac{\ln(1+x)}{x} = F(1, 1, 2; -x), \qquad \frac{\ln(1-x)}{(-x)} = F(1, 1, 2; x)$$

３．超幾何関数のＪプログラム

ここではＪではあるが、在来型のループを使ってプログラムを作った。

```
Fn =: 3 : 0
:
n =. 10
'a b c' =. x.
x =. y.
i =. 1
A =. a
B =. b
C =. c
s =. 1
while. i < n
  do.
      NB. calc. coeficient ==============================
NB.     wr  (2":  i),  ':',  (3":A),  '<',  (":*/A),  '>  *',  (3":B),  '<',
(":*/B),'> /',(3":C),'<', (":*/C),'>'
      CC =. ((,*/A) * (,*/B)) % (,*/C)
NB.   wr 'CC=', (": CC), ' : ', (": x: CC)
NB.   wr 's=', ":s
      A =. A, (a+i)
      B =. B, (b+i)
      C =. C, (c+i)
      NB. calc. x-value =================================
      Xi =. (% ! i) * (x^i)
      sx =. CC * Xi
      s =. s + sx
      i =. i + 1
  end.
s
)
```

まずは、超幾何関数のＪプログラムFnの途中の計算結果を示しつつ、実行してみる。

```
 (1, 1, 2) Fn 0.5
 1:  1<1> *  1<1> /  2<2>
CC=0.5 : 1r2
s=0
xi=0.5
sx=0.25

 2:  1  2<2> *  1  2<2> /  2  3<6>
CC=0.666667 : 2r3
s=0.25
xi=0.125
sx=0.0833333

 3:  1  2  3<6> *  1  2  3<6> /  2  3  4<24>
CC=1.5 : 3r2
s=0.333333
xi=0.0208333
sx=0.03125

 4:  1  2  3  4<24> *  1  2  3  4<24> /  2  3  4  5<120>
CC=4.8 : 24r5
s=0.364583
xi=0.00260417
sx=0.0125

 5:  1  2  3  4  5<120> *  1  2  3  4  5<120> /  2  3  4  5  6<720>
CC=20 : 20
s=0.377083
xi=0.000260417
sx=0.00520833

（途中省略）

 9:  1  2  3  4  5  6  7  8  9<362880> *  1  2  3  4  5  6  7  8  9<362880> /
2  3  4  5  6  7  8  9 10<3628800>
```

CC=36288 : 36288

s=0.385934

xi=5.38229e_9

sx=0.000195313

1.38613

次は通常のとおり計算結果のみを出力する。そして自然対数の値と比較する。

(1, 1, 2) Fn 0.5 ======== F (1, 1, 2; 0.5) に相当

1.38613

(^. (1 - 0.5)) % (_0.5) === ln (1 - 0.5) % (-0.5)の計算値

1.38629

(1, 1, 2) Fn _0.5 ======== F (1, 1, 2; -0.5) に相当

0.810869

(^. (1 + 0.5)) % 0.5 === ln (1 + 0.5) % (0.5)の計算値

0.81093

このようにして、超幾何関数のプログラムFnにより、自然対数の値が得られる。

４．合流型超幾何関数のＪプログラムとその利用

Gaussの超幾何関数とは、少し違ってKummerの超幾何関数(合流形超幾何関数)というのがある。先のFnを少し手直して、Mnとしてつくった。定義は巻末にあげた。

実は、これを用いることによって、確率・統計などの基本である、誤差関数(Error Function)や正規分布の累積関数(Normal Cummulative Distribution Function)などが作られる。

$$erf\,(x) = \frac{2}{\sqrt{\pi}}\int_0^x \exp(-t^2)dt = \frac{2x\exp(-x^2)}{\sqrt{\pi}}M(1,1.5;\,x^2)$$

$$ncd\,(x) = \frac{1}{\sqrt{2\pi}}\int_{-\infty}^x \exp(-t^2/2)\,dt$$

式の詳細は「岩波数学公式Ⅲ」p.57あるいは以下のホームページの記述を参照のこと

Wolfram Mathworld Erf Error Function

Ｊのプログラムと実行はつぎのようになる。

誤差関数は次のように求められる。

```
NB. Error Function / T. N version 1 ============
Merf =: 3 : 0
M1 =. (1 1.5) Mn *: y.
M2 =. ((2 % %: 1p1) * y.) % (^ *: y.)
M =. M1 * M2
)
   (1, 1.5) Mn 0.5
1.41069
   (1, 1.5) Mn 1
2.03008
   , Merf"(0) 0.5 1 1.5
0.5205 0.842701 0.96605
```

正規分布の－∞から各値までの累積値は次のように計算される。

```
NB. Cumulative Normal Function 0-1 / T.N =============
CumNormf =: 3 : 0
0.5 * 1 + Merf (%: 0.5) * y.
)
   , CumNormf"(0) _2 _1.5 _1 _0.5 0 0.5 1 1.5 2
0.0227601 0.0668073 0.158655 0.308538 0.5 0.691462 0.841345 0.933193 0.97724
```

５．終わりにー付記

今回の発表は、ＪのProgram ManualにH. Hypergeometric Functionとして以下のような記述があるが、tacit形式で副詞、接続詞を用いた、まさにトリッキーなコーディングである。実は、これをフォローしようとしたが、とても一筋縄では行かなかった。

結局、数学の原理に基づき、自分流にExplicitでループを使用して定義し、それにより超幾何関数なるものをやっと理解することが出来た（計算の数値の照合にはおおいに利用したが）というのが本音である。

```
NB. Hypergeometric Function - J Original Version ==================
   rf=: 1 : '(,x.)"_ ^!.1/ i.@['
   L1=: 2 : 'x.rf %&(*/) y.rf'
   L2=: (i.@[ ^~ ]) % (!@i.@[)
   H=: L1 (+/ . *) L2

   erf=: 1 H. 1.5@*: * 2p_0.5&* % ^@:*:
   n01cdf =: -: @: >: @: erf @: ((%:0.5)&*)
```

Jプログラムリスト

```
NB. Hypergeometric Function ==================================
NB. Gauss Hypergeometric Function
NB. test natural logarithm calc.

NB.   (1, 1, 2) Fn _0.5       for F(1, 1, 2;_0.5)
NB. 0.810869
NB.
NB.   (^. (1 + 0.5)) % 0.5    for ln(1 + 0.5) % (0.5)
NB. 0.81093

NB.   (1, 1, 2) Fn 0.5        for F(1, 1, 2;0.5)
NB. 1.38613
NB.   (^. (1 - 0.5)) % (_0.5) for ln(1 - 0.5) % (_0.5)
NB. 1.38629

Fn =: 3 : 0
:
n =. 10
'a b c' =. x.
x =. y.
i =. 1
A =. a
B =. b
C =. c
s =. 1
while. i < n
  do.
     NB. calc. coeficient ==============================
NB.    wr  (2": i),  ':',  (3":A),  '<',  (":*/A),  '> *',  (3":B),  '<',
(":*/B),'> /',(3":C),'<', (":*/C),'>'
     CC =. ((,*/A) * (,*/B)) % (,*/C)
NB.  wr 'CC=', (": CC), ' : ', (": x: CC)
NB.  wr 's=', ":s
     A =. A, (a+i)
     B =. B, (b+i)
```

```
     C =. C, (c+i)
     NB. calc. x-value ================================
     Xi =. (% ! i) * (x^i)
     sx =. CC * Xi
     s =. s + sx
     i =. i + 1
  end.
s
)

NB. Kummer Cofluent Hypergeometric Function ============
NB.   (1, 1.5) Mn 0.5
NB. 1.41069
NB.   (1, 1.5) Mn 1
NB. 2.03008

Mn =: 3 : 0
:
n =. 10
'a c' =. x.
x =. y.
i =. 1
A =. a
B =. 1
C =. c
s =. 1
while. i < n
  do.
     NB. calc. coeficient
NB.    wr  (2":  i),  ':',  (3":A),  '<',  (":*/A),  '>  *',  (3":B),  '<',
(":*/B),'> /',(3":C),'<', (":*/C),'>'
     CC =. ((,*/A) * (,*/B)) % (,*/C)
NB.  wr 'CC=', (": CC), ' : ', (": x: CC)
NB.  wr 's=', ":s
     A =. A, (a+i)
     B =. B, 1  NB. for Kummer Function
```

```
NB.     B =. B, (b+i)
     C =. C, (c+i)
     NB. calc. x-value
     Xi =. (% ! i) * (x^i)
     sx =. CC * Xi
     s =. s + sx
     i =. i + 1
  end.
s
)

NB. error function ============================
adj_erf =: 2p_0.5&* % ^ @: *:

NB. J original ===================
f1 =: (1 H. 1.5)@*:
ferf =: f1 * adj_erf
NB.    ferf 0.5 1
NB. 0.5205 0.842701

NB. Nishikawa version 0 ============
m0 =: 3 : '(1, 1.5) Mn (*: y.)'
merf =: m0 * adj_erf
NB.    , merf"(0) 0.5 1
NB. 0.5205 0.842701
NB. Nishikawa version 1 ============
Merf =: 3 : 0
 M1 =. (1 1.5) Mn *: y.
 M2 =. ((2 % %: 1p1) * y.) % (^ *: y.)
 M =. M1 * M2
)

NB. Cumulative Normal Function 0-1 =============
CumNormf =: 3 : 0
0.5 * 1 + Merf (%: 0.5) * y.
)
```